

PSEINT

PRÁCTICA II



PSEINT - PRÁCTICA II

INSTRUCCIONES PREVIAS

Ahora que has aprendido los conceptos básicos de PSeInt, toca ponerlos en práctica...¡No tiene misterio!

Sigue los siguientes pasos, practica, y aplica así todo lo aprendido, comprobando si lo estás haciendo correctamente.

Para comenzar, te recomendamos seguir los siguientes pasos:

- **Instala la herramienta PSeInt** que encontrarás en la sección **contenido descargable**. Ahí encontrarás el **link directo** a la **"Descarga de PSeInt"**.
- **Configura la herramienta** seleccionando el Lenguaje Flexible, será el que utilizemos siempre. Para ello:

[Configurar → Opciones del Lenguaje (perfiles) → Flexible → Aceptar]

¡Todo preparado!, continua con los siguientes pasos:

- **Crea tu algoritmo**, personalízalo con el nombre que consideres. De esta forma, con un simple vistazo, sabrás qué proceso tratará.
- **Construye tu algoritmo**, ponte manos a la obra. Lee con atención el enunciado de cada actividad.
- **Ejecuta tu algoritmo**, una vez has completado tu algoritmo con todas las instrucciones (pasos), haz **click** en el icono verde que indica **"Ejecutar"**.
- **Prueba tu algoritmo**, al final de este documento encontrarás las soluciones a las actividades.

Ten en cuenta que algunas de las instrucciones que se proporcionan en las soluciones puedes no haberlas visto en el curso teórico, pero tienes los conocimientos suficientes para investigar y buscar esta u otras soluciones.

Recuerda que en el mundo de la informática, el cuál evoluciona tan rápidamente, es una práctica muy común tener que investigar y autoformarte, a través de diversas fuentes, para ampliar tus conocimientos.

PSEINT - PRÁCTICA II

ACTIVIDADES

Actividad 1. Hacer un programa que pida un nombre y muestre como salida: "Hola XXXXX has creado tu primer programa".

Actividad 2. Hacer un programa que debe pedir un nombre y la edad de una persona y muestre como salida: "Hola XXXXX tienes YY años".

Actividad 3. Crear el pseudocódigo para un programa o proceso que cree dos variables numéricas, que les asigne un valor inicial de 5 y 7 respectivamente. Posteriormente, hacer la suma de dichas variables guardando el resultado en una tercera variable y mostrar en pantalla el resultado.

Actividad 4. Pedir dos números, y realizar las operaciones algebraicas de suma, resta, multiplicación, división, resto, y potencia. Muestra los resultados por pantalla.

Actividad 5. Hacer un algoritmo para calcular el área de un cuadrado, donde introduciremos la longitud de uno de sus lados.

Actividad 6. Hacer un algoritmo donde se calcule la hipotenusa de un triángulo, donde introduciremos las longitudes de los catetos.

Actividad 7. Calcular el área de un círculo, donde introduciremos el radio de esta. Usaremos la variable predefinida de PseInt PI.

Actividad 8. Crear un programa donde se pidan dos números, que usaremos para realizar comparaciones con los siguientes operadores relacionales.

Ej. a = 5; b = 7;

a) a = b;	f) b=a;
b) a<b;	g) b<a;
c) a>b;	h) b>a;
d) a>=b;	i) b<=a;
e) a<=b;	j) b<=a;

Y mostraremos el resultado de cada una de las comparaciones.

Actividad 9. Crear un programa, que lea un número y diga si está dentro del rango de 0 a 10, ambos inclusive.

Actividad 10. Crear un programa que lea un número y diga que es cierto si es mayor de 20 o es impar.

Actividad 11. Algoritmo que muestre por pantalla el número mayor de 2 números que introducirá el usuario.

Actividad 12. Algoritmo que haga la suma de los números, solo si los números son mayores de 5, si alguno de los dos números es menor, que muestre un mensaje de texto indicando que no se puede hacer la operación.

Actividad 13. Algoritmo que haga la multiplicación de dos números que introducirá el usuario. Si el resultado de la multiplicación es mayor de 100 mostrará el mensaje "Mayor de 100", en caso contrario "Menor de 100", y si es 100 mostrará "El resultado es 100".

Actividad 14. Usando la estructura "Segun", leer un número y mostrar un mensaje de si es par o impar.

Actividad 15. Crear programa, donde introducimos dos números y después nos sale un menú, indicando la operación matemática que queremos realizar, mostrando finalmente el resultado de esta, y en caso de seleccionar una opción que no este contemplado, mostrar un mensaje de opción no valida.

Actividad 16. Crear un programa, donde leerá un número y hará una iteración hasta 100, mostrándolo en pantalla y, sumando sus valores donde finalmente mostrará el resultado.

Actividad 17. Hacer una iteración de 100 a 0, ambos inclusive, y mostrarlo en pantalla.

Actividad 18. Hacer el mismo ejercicio que el 14, pero usando la estructura Repetir. Observaremos la diferencia entre uno y otro al poner un valor por encima de 100.

Actividad 19. Crea un algoritmo que leerá texto pasado por la pantalla, continuamente hasta que escribamos "Fin". Todo el texto se irá concatenando. Al finalizar el ciclo, mostrará el texto final.

Actividad 20. Introducir un texto, y un número. Escribir el texto tantas veces como se ha indicado en el número.

Actividad 21. Escribir los números del 0 al 20 de manera decreciente.

Actividad 22. Obtener la media de 5 números leídos por teclado.

Actividad 23. Hacer un programa donde se pasará un número, dirá si es par o impar, y al final preguntará si quiere introducir un nuevo número.

Actividad 24. Imprimir en pantalla una matriz de 10 X 10 con el carácter “*”.

Para este ejercicio se ha de usar la función “Escribir Sin Saltar”

Actividad 25. Crear un programa que sea una calculadora con un menú donde se podrá seleccionar que tipo de operación se quiere realizar, además de una última opción que será para salir de programa. Mientras no se seleccione “Salir”, el programa se repetirá.

Actividad 26. Modularizar la actividad anterior.

Actividad 27. Se tiene el siguiente menú de un restaurante de bocadillos. Diseñar un algoritmo capaz de leer el número de unidades consumidas de cada alimento ordenado y calcular la cuenta total. Vamos a suponer que estos precios son fijos, es decir, que son constantes.

Recuerda que en PSeInt no se usa comas para separar la parte decimal de la parte entera.

PRODUCTO	PRECIO
Bocadillo de jamón	1,5 €
Refresco	1,05 €
Cerveza	0,75 €

Actividad 28. Sobre el importe introducido por el usuario, calcular el IVA (21%). Mostrar cuanto se incrementa el precio, y el precio total con el IVA.

Actividad 29. Hacer un pequeño juego donde el usuario tendrá que adivinar un número generado por la aplicación entre 0 y 100. Tendrá 10 intentos, en cada intento si no lo ha acertado se le indicará al usuario un mensaje de si el número es mayor o menor con respecto al que tiene que acertar.

Actividad 30. Calcular la nota de N alumnos, introduciendo su nota teórica (60%) y su nota practica (40%). Mostrarlo por pantalla.

PSEINT - PRÁCTICA II

SOLUCIONES A LAS ACTIVIDADES

Actividad 1. Hacer un programa que pida un nombre y muestre como salida: "Hola XXXXX has creado tu primer programa".

```
Proceso Ejercicio1
    Definir nombre Como Cadena;
    Leer nombre;
    Escribir "Hola ",nombre," has creado tu primer programa.";
FinProceso
```

Actividad 2. Hacer un programa que debe pedir un nombre y la edad de una persona y muestre como salida: "Hola XXXXX tienes YY años".

```
Proceso Ejercicio2
    Definir nombre,edad Como Cadena;
    Leer nombre;
    Leer edad;
    Escribir "Hola ",nombre," tienes ",edad," años.";
FinProceso
```

Actividad 3. Crear el pseudocódigo para un programa o proceso que cree dos variables numéricas, que les asigne un valor inicial de 5 y 7 respectivamente. Posteriormente, hacer la suma de dichas variables guardando el resultado en una tercera variable y mostrar en pantalla el resultado.

```
Proceso Ejercicio3
    Definir num1,num2,resultado Como Entero
    num1 = 5
    num2 = 7
    resultado = num1+num2
    Escribir 'El resultado de la suma de las dos variables es : ',resultado
FinProceso
```

Actividad 4. Pedir dos números, y realizar las operaciones algebraicas de suma, resta, multiplicación, división, resto, y potencia. Muestra los resultados por pantalla.

```
Proceso Ejercicio4
    Definir num1, num2, resultado Como Real;
    Escribir "Introduce el primer número: ";
    Leer num1;
    Escribir "Introduce el segundo número: ";
    Leer num2;
    resultado=num1+num2;
    Escribir "Suma: ", resultado;
    resultado=num1-num2;
    Escribir "Resta: ", resultado;
    resultado=num1*num2;
    Escribir "Multiplicación: ",resultado;
    resultado=num1/num2;
    Escribir "División: ",resultado;
    resultado=num1%num2;
    Escribir "Resto: ",resultado;
    resultado=num1^num2;
    Escribir "Potenciación: ", resultado;
FinProceso
```

Actividad 5. Hacer un algoritmo para calcular el área de un cuadrado, donde introduciremos la longitud de uno de sus lados.

```
Proceso Ejercicio5
    Definir lado, area Como Entero;
    Escribir "Introduce la longitud de uno de los lados del cuadrado: ";
    Leer lado;
    area= lado^2;
    Escribir "El area del cuadrado es: ",area;
FinProceso
```

Actividad 6. Hacer un algoritmo donde se calcule la hipotenusa de un triángulo, donde introduciremos las longitudes de los catetos.

```
Proceso Ejercicio6
    Definir cateto1, cateto2, hipotenusa Como Real;
    Escribir "Introduce la longitud del primer cateto: ";
    Leer cateto1;
    Escribir "Introduce la longitud del segundo cateto: ";
    Leer cateto2;
    hipotenusa = ((cateto1^2) + (cateto2^2))^(1/2);
    Escribir "La hipotenusa de los catetos ",cateto1, " y ",cateto2, " es ",hipotenusa;
FinProceso
```


Actividad 7. Calcular el área de una círculo, donde introduciremos el radio de esta. Usaremos la variable predefinida de PseInt PI.

```
Proceso Ejercicio7
    Definir radio,area Como Real;
    Escribir "Introducir la longitud del radio: ";
    Leer radio;
    area = PI*radio^2;
    Escribir "El area del círculo es ", area;
FinProceso
```

Actividad 8. Crear un programa donde se pidan dos números, que usaremos para realizar comparaciones con los siguientes operadores relacionales.

Ej. a = 5; b = 7;

 a) a = b; f)b=a;

 b) a<b; g)b<a;

 c) a>b; h)b>a;

 d)a>=b; i) b<=a;

 e)a<=b; j)b<=a;

Y mostraremos el resultado de cada una de las comparaciones.

```
Proceso Ejercicio8
    Definir num1, num2 Como Caracter;
    Definir res Como Logico;
    Escribir "Introduce dos números: ";
    Leer num1, num2;
    Escribir num1 ">", num2 "=", num1>num2;
    Escribir num1 "<", num2 "=", num1<num2;
    Escribir num1 "=", num2 "=", num1=num2;
    Escribir num1 "<=", num2 "=", num1<=num2;
    Escribir num1 ">=", num2 "=", num1>=num2;
FinProceso
```

Actividad 9. Crear un programa, que lea un número y diga si está dentro del rango de 0 a 10, ambos inclusive.

```
Proceso Ejercicio9
    Definir num Como Entero;
    Definir res Como Logico;
    Escribir "Introduce un número: ";
    Leer num;
    res = num>=0 & num<=10;
    Escribir "El num es ", res, " que está dentro del rango de 0 a 10";
FinProceso
```

Actividad 10. Crear un programa que lea un número y diga que es cierto si es mayor de 20 o es impar.

```
Proceso Ejercicio10
    Definir num,impar Como Entero;
    Definir res Como Logico;
    Escribir "Introduce un número: ";
    Leer num;
    impar = num MOD 2;
    res = num>20 | impar=1;
    Escribir "El num es ", res, " que es impar o mayor de 20";
FinProceso
```

Actividad 11. Algoritmo que muestre por pantalla el número mayor de 2 números que introducirá el usuario.

```
Proceso Ejercicio11
    Definir num1,num2 Como Entero;
    Escribir "Introduce dos números: ";
    Leer num1;
    Leer num2;
    Si (num1>num2) Entonces
        Escribir "El numero mayor es ",num1;
    SiNo
        Escribir "El numero mayor es ",num2;
    FinSi
FinProceso
```

Actividad 12. Algoritmo que haga la suma de los números solo si los números son mayores de 5. Si alguno de los dos números es menor, se mostrará un mensaje de texto indicando que no se puede hacer la operación.

```
Proceso Ejercicio12
    Definir num1, num2 Como Entero;
    Leer num1;
    Leer num2;
    Si num1>5 Entonces
        Si num2>5 Entonces
            Escribir num1+num2;
        SiNo
            Escribir num2, " es menor de 5.";
        FinSi
    SiNo
        Escribir num1, " es menor de 5";
    FinSi
FinProceso
```

Actividad 13. Algoritmo que haga la multiplicación de dos números que introducirá el usuario. Si el resultado de la multiplicación es mayor de 100 mostrará el mensaje "Mayor de 100", en caso contrario "Menor de 100", y si es 100 mostrará "El resultado es 100".

```
Proceso Ejercicio13
    Definir num1,num2,resultado Como Entero;
    Escribir "Introduce dos números: ";
    Leer num1;
    Leer num2;
    resultado = num1 * num2;
    Si resultado =100 Entonces
        Escribir "El resultado es 100";
    SiNo
        Si resultado <100 Entonces
            Escribir "El resultado es menor de 100";
        SiNo
            Escribir "El resultado es mayor de 100";
        FinSi
    FinSi
FinProceso
```

Actividad 14. Usando la estructura “Segun”, leer un número y mostrar un mensaje de si es par o impar.

```
Proceso Ejercicio14
    Definir num, resultado Como Entero;
    Escribir "Introduce un número: ";
    Leer num;
    resultado = num MOD 2;
    Segun resultado Hacer
        0: Escribir "El número es PAR";
        1: Escribir "El numero es IMPAR";
    FinSegun
FinProceso
```

Actividad 15. Crear programa, donde introducimos dos números y después nos sale un menú, indicando la operación matemática que queremos realizar, mostrando finalmente el resultado de esta, y en caso de seleccionar una opción que no este contemplado, mostrar un mensaje de opción no valida.

```
Proceso Ejercicio15
    Definir num1,num2, opcion Como Entero;
    Definir resultado como real;
    Escribir "Introduce los dos números: ";
    Leer num1;
    Leer num2;
    Escribir "Selecciona la operación que quieres realizar: ";
    Escribir "1. Suma";
    Escribir "2. Resta";
    Escribir "3.Multiplicación";
    Escribir "4. División";
    Escribir "Opción: ";
    Leer opcion;
    Segun opcion Hacer
        1: resultado = num1+num2
            Escribir "El resultado de la suma es: ", resultado
        2: resultado = num1-num2
            Escribir "El resultado de la resta es: ", resultado
        3: resultado = num1*num2
            Escribir "El resultado de la multiplicacion es: ", resultado
        4: resultado = num1/num2
            Escribir "El resultado de la division es: ", resultado
        De Otro Modo:
            Escribir "No has seleccionado una opción válida.";
    FinSegun
FinProceso
```

Actividad 16. Crear un programa, donde leerá un número y hará una iteración hasta 100, mostrándolo en pantalla y, sumando sus valores donde finalmente mostrará el resultado.

```
Proceso Ejercicio16
    Definir num,resultado Como Entero;
    Leer num;
    resultado=0;
    Mientras num<100 Hacer
        Escribir "Número: ",num;
        num = num+1;
        resultado = resultado +num;
    FinMientras
    Escribir "El resultado final de la suma es: ", resultado;
FinProceso
```

Actividad 17. Hacer una iteración de 100 a 0, ambos inclusive, y mostrarlo en pantalla.

```
Proceso Ejercicio17
    Definir contador Como Entero;
    contador = 100;
    Mientras (contador>=0) Hacer
        Escribir contador;
        contador = contador -1;
    FinMientras
FinProceso
```

Actividad 18. Hacer el mismo ejercicio que la actividad 16, pero usando la estructura Repetir. Observaremos la diferencia entre uno y otro al poner un valor por encima de 100.

```
Proceso Ejercicio18
    Definir num,resultado Como Entero;
    Leer num;
    resultado=0;
    Repetir
        Escribir "Número: ",num;
        num = num+1;
        resultado = resultado +num;
    Hasta que num>=100
    Escribir "El resultado final de la suma es: ", resultado;
FinProceso
```

Actividad 19. Crea un algoritmo que leerá texto pasado por la pantalla, continuamente hasta que escribamos "Fin". Todo el texto se irá concatenando. Al finalizar el ciclo, mostrará el texto final.

```
Proceso Ejercicio19
    Definir letras,cadenaCaracteres Como Cadena;

    cadenaCaracteres = "";
    Repetir
        Leer letras;
        cadenaCaracteres = cadenaCaracteres + letras;
    Hasta Que letras ="Fin"
    Escribir cadenaCaracteres;
FinProceso
```

Actividad 20. Introducir un texto, y un número. Escribir el texto tantas veces como se ha indicado en el número.

```
Proceso Ejercicio20
    Definir mensaje Como Cadena;
    Definir num, contador Como Entero;
    Leer mensaje;
    Leer num;

    Para contador=1 Hasta num Hacer
        Escribir mensaje;
    FinPara
FinProceso
```

Actividad 21. Escribir los números del 0 al 20 de manera decreciente.

```
Proceso Ejercicio21
    Definir contador Como Entero;

    Para contador=20 Hasta 0 con Paso -1 Hacer
        Escribir contador;
    FinPara
FinProceso
```

Actividad 22. Obtener la media de 5 números leídos por teclado.

```
Proceso Ejercicio22
    Definir contador Como Entero;
    Definir total, num Como Real;
    total = 0;
    Para contador=1 Hasta 5 Hacer
        Leer num;
        total = total + num;
    FinPara
    total = total / 5;
    Escribir "La media de los 5 números es ", total;
FinProceso
```

Actividad 23. Hacer un programa donde se pasará un número, dirá si es par o impar, y al final preguntará si quiere introducir un nuevo número.

```
Proceso Ejercicio23
    Definir num Como Entero;
    Definir letra Como Caracter;
    Definir continuar Como Logico;
    continuar=Verdadero;
    Mientras (continuar) Hacer
        Escribir "Introduce un número: ";
        Leer num;
        Si (num MOD 2 = 0) Entonces
            Escribir "El numero ", num, " es PAR";
        SiNo
            Escribir "El numero ", num, " es IMPAR";
        FinSi
        Escribir "Quieres introducir otro número (S/N): ";
        Leer letra;
        Si letra="N" o letra="n" Entonces
            continuar=Falso;
        FinSi
    FinMientras
FinProceso
```

Actividad 24. Imprimir en pantalla una matriz de 10 X 10 con el carácter "*".

```
Proceso Ejercicio24
    Definir filas, columnas Como Entero;
    Para filas =1 Hasta 10 Hacer
        Para columnas = 1 Hasta 10 Hacer
            Escribir "*" Sin Saltar;
        FinPara
        Escribir "";
    FinPara
FinProceso
```

Actividad 25. Crear un programa que sea una calculadora con un menú donde se podrá seleccionar que tipo de operación se quiere realizar, además de una última opción que será para salir de programa. Mientras no se seleccione "Salir", el programa se repetirá.

```
Proceso Ejercicio25
    Definir resultado Como Real;
    Definir num1, num2, opcion Como Entero;
    Definir continuar Como Logico;
    continuar=Verdadero;
    Mientras continuar Hacer
        Escribir "Selecciona la operación que quieres realizar: ";
        Escribir "1. Suma";
        Escribir "2. Resta";
        Escribir "3. Multiplicación";
        Escribir "4. División";
        Escribir "5. Salir";
        Escribir "Opción: ";
        Leer opcion;
        Escribir "Introduce los dos números: ";
        Leer num1;
        Leer num2;
        Segun opcion Hacer
            1: resultado = num1 + num2;
            2: resultado = num1 - num2;
            3: resultado = num1 * num2;
            4: resultado = num1 / num2;
            5: continuar = Falso;
        De Otro Modo:
            Escribir "No has seleccionado una opción válida.";
        FinSegun
        Si opcion=1 o opcion=2 o opcion=3 o opcion=4 entonces
            Escribir resultado;
        FinSi
    FinMientras
FinProceso
```


Actividad 26. Modularizar la actividad anterior.

```
SubProceso resultado <- Suma ()
    Definir num1, num2,resultado Como Entero;
    Escribir "Introduce los dos números: ";
    Leer num1;
    Leer num2;
    resultado = num1+num2;
FinSubProceso
SubProceso resultado <- Resta ()
    Definir num1, num2,resultado Como Entero;
    Escribir "Introduce los dos números: ";
    Leer num1;
    Leer num2;
    resultado = num1-num2;
FinSubProceso
SubProceso resultado <- Multiplicacion ()
    Definir num1, num2,resultado Como Entero;
    Escribir "Introduce los dos números: ";
    Leer num1;
    Leer num2;
    resultado = num1*num2;
FinSubProceso
SubProceso resultado <- Division ()
    Definir num1, num2 Como Entero;
    Definir resultado Como Real;
    Escribir "Introduce los dos números: ";
    Leer num1;
    Leer num2;
    resultado = num1/num2;
FinSubProceso

SubProceso opcion <- Menu ()
    Definir opcion Como Entero;
    Escribir "Selecciona la operación que quieres realizar: ";
    Escribir "1. Suma";
    Escribir "2. Resta";
    Escribir "3. Multiplicación";
    Escribir "4. División";
    Escribir "5. Salir";
    Escribir "Opción: ";
    Leer opcion;

FinSubProceso
```

Proceso Ejercicio26

```
Definir resultado Como Real;
Definir opcion Como Entero;
Definir continuar Como Logico;
continuar=Verdadero;
Mientras continuar Hacer
    opcion <-Menu();
    Segun opcion Hacer
        1: resultado = Suma();
        2: resultado = Resta();
        3: resultado = Multiplicacion();
        4: resultado = Division();
        5: continuar = Falso;
    De Otro Modo:
        Escribir "No has seleccionado una opción válida.";
    FinSegun
    Si opcion=1 o opcion=2 o opcion=3 o opcion=4 entonces
        Escribir resultado;
    FinSi
FinMientras
FinProceso
```

Actividad 27. Se tiene el siguiente menú de un restaurante de bocadillos. Diseñar un algoritmo capaz de leer el número de unidades consumidas de cada alimento ordenado y calcular la cuenta total. Vamos a suponer que estos precios son fijos, es decir, que son constantes. Recuerda que en PSeInt no se usa comas para separar la parte decimal de la parte entera.

PRODUCTO	PRECIO
Bocadillo de jamón	1,5 €
Refresco	1,05 €
Cerveza	0,75 €

Proceso ejercicio27

```
JAMON=1.5
REFRESCO=1.05
CERVEZA=0.75
total=0
Escribir "Introduce la cantidad de bocadillos de jamon"
Leer cant_jamon
Escribir "Introduce la cantidad de refresco"
Leer cant_refresco
Escribir "Introduce la cantidad de cerveza"
Leer cant_cerveza
//NOTA: los dos siguientes van en una linea
total<-((cant_jamon*JAMON)+(cant_refresco*REFRESCO)
+(cant_cerveza*CERVEZA))
Escribir total
FinProceso
```

Actividad 28. Sobre el importe introducido por el usuario, calcular el IVA (21%). Mostrar cuanto se incrementa el precio, y el precio total con el IVA.

```
Proceso Practica28
  //Definimos unas variables
  Definir importe, incremento, total Como Real

  //Solicitamos y guardamos el valor sobre el cual aplicar el IVA
  Escribir Sin Saltar "Introduce el importe sobre el cual aplicar el IVA (21%): ";
  Leer importe;

  //Ahora calculamos el incremento y el total para mostrarlo de nuevo
  incremento <- importe * 0.21;
  total <- importe + incremento;

  //Mostramos el resultado obtenido
  Escribir "Incremento del IVA: ", incremento;
  Escribir "Precio Total con IVA: ", total;

FinProceso
```

Actividad 29. Hacer un pequeño juego donde el usuario tendrá que adivinar un número generado por la aplicación entre 0 y 100. Tendrá 10 intentos, en cada intento si no lo ha acertado se le indicará al usuario un mensaje de si el número es mayor o menor con respecto al que tiene que acertar.

```
Proceso Practica29
  //Definimos las variables a usar
  Definir numAzar, num, intentos Como Entero;

  //Calculamos el número al azar que debe adivinar el usuario
  numAzar <- AZAR(100);
  intentos <- 1;
  //Solicitamos un número al usuario y sumamos los intentos
  Repetir
    Escribir "Intento",intentos,". Introduce un número: ";
    Leer num;
    Si (num = numAzar) Entonces
      Escribir "Correcto, has acertado el número!!";
      intentos <- 10;
    Sino
      Si (num < numAzar) Entonces
        Escribir "Incorrecto, el número introducido es menor.";
      Sino
        Escribir "Incorrecto, el número introducido es mayor.";

      FinSi
      intentos <- intentos + 1;
    FinSi
  Hasta Que intentos = 11;

FinProceso
```

Actividad 30. Calcular la nota de N alumnos, introduciendo su nota teórica (60%) y su nota practica (40%). Mostrarlo por pantalla.

```
Proceso Ejercicio30
    //Validamos el numero de alumnos
    Repetir
        Escribir "Escribe el numero de alumnos"
        leer alumnos

        Si alumnos<1 Entonces
            escribir "Debe ser mayor o igual que 1"
        Fin Si

    Hasta Que alumnos>0

    //Variables que vamos a usar
    nota_teorica = 0
    nota_practica = 0
    nota = 0
    //Recorremos el numero de alumnos
    Para i<-1 Hasta alumnos Con Paso 1 Hacer

        //Validamos la nota teórica
        Repetir

            Escribir "Introduce la nota teórica del alumno ",i
            leer nota_teorica
            Si no (nota_teorica>=0 y nota_teorica<=10) Entonces
                escribir "Debes escribir un valor entre 0 y 10"
            Fin Si

        Hasta Que nota_teorica>=0 y nota_teorica<=10

        //Validamos la nota practica
        Repetir

            Escribir "Introduce la nota practica del alumno ",i
            leer nota_practica
            Si no (nota_practica>=0 y nota_practica<=10) Entonces
                escribir "Debes escribir un valor entre 0 y 10"
            Fin Si

        Hasta Que nota_practica>=0 y nota_practica<=10
        //Calculamos la nota
        nota = (nota_teorica*0.6) + (nota_practica*0.4)
        //mostramos la nota
        Escribir "El alumno numero ",i," ha sacado una nota de ", nota

    Fin Para
FinProceso
```

**FUNDACIÓN
ACCENTURE**

accenture